Rules for integrands of the form $(a + b\,\mathrm{Tan}[c + d\,x])^n$

1. $\int (b\,\mathrm{Tan}[c + d\,x])^n \,dx$

   1: $\int (b\,\mathrm{Tan}[c + d\,x])^n \,dx$ when $n > 1$

- **Reference: G&R 2.510.1, CRC 423, A&S 4.3.129**
- **Reference: G&R 2.510.4, CRC 427, A&S 4.3.130**
- **Derivation: Algebraic expansion**
- **Basis:** $(b\,\mathrm{Tan}[z])^n == b\,b\,\mathrm{Sec}[z]^2\,(b\,\mathrm{Tan}[z])^{n-2} - b^2\,(b\,\mathrm{Tan}[z])^{n-2}$
- **Rule: If** $n > 1$**, then**

$$\int (b\,\mathrm{Tan}[c + d\,x])^n \,dx \longrightarrow \frac{b\,(b\,\mathrm{Tan}[c + d\,x])^{n-1}}{d\,(n-1)} - b^2 \int (b\,\mathrm{Tan}[c + d\,x])^{n-2} \,dx$$

- **Program code:**

```
Int[(b_.*tan[c_.+d_.*x_])^n_,x_Symbol] :=
  b*(b*Tan[c+d*x])^(n-1)/(d*(n-1)) -
  b^2*Int[(b*Tan[c+d*x])^(n-2),x] /;
FreeQ[{b,c,d},x] && GtQ[n,1]
```

2: $\int (b\,\text{Tan}[c+d\,x])^n\,dx$ when $n<-1$

- **Reference: G&R 2.510.4, CRC 427'**
- **Reference: G&R 2.510.1, CRC 423'**
- **Derivation: Algebraic expansion**
- **Basis:** $(b\,\text{Tan}[z])^n == \text{Sec}[z]^2\,(b\,\text{Tan}[z])^n - \frac{1}{b^2}\,(b\,\text{Tan}[z])^{n+2}$
- **Rule: If** $n<-1$**, then**

$$\int (b\,\text{Tan}[c+d\,x])^n\,dx \longrightarrow \frac{(b\,\text{Tan}[c+d\,x])^{n+1}}{b\,d\,(n+1)} - \frac{1}{b^2}\int (b\,\text{Tan}[c+d\,x])^{n+2}\,dx$$

- **Program code:**

```
Int[(b_.*tan[c_.+d_.*x_])^n_,x_Symbol] :=
  (b*Tan[c+d*x])^(n+1)/(b*d*(n+1)) -
  1/b^2*Int[(b*Tan[c+d*x])^(n+2),x] /;
FreeQ[{b,c,d},x] && LtQ[n,-1]
```

3: $\int \text{Tan}[c+d\,x]\,dx$

Reference: G&R 2.526.17, CRC 292, A&S 4.3.115

- **Reference: G&R 2.526.33, CRC 293, A&S 4.3.118**
- **Derivation: Integration by substitution**
- **Basis:** $\text{Tan}[c+d\,x] == -\frac{1}{d\,\text{Cos}[c+d\,x]}\,\partial_x \text{Cos}[c+d\,x]$
- **Rule:**

$$\int \text{Tan}[c+d\,x]\,dx \longrightarrow -\frac{\text{Log}[\text{Cos}[c+d\,x]]}{d}$$

- **Program code:**

```
Int[tan[c_.+d_.*x_],x_Symbol] :=
  -Log[RemoveContent[Cos[c+d*x],x]]/d /;
FreeQ[{c,d},x]
```

x: $\int \frac{1}{\text{Tan}[c + d\,x]} \, dx$

Note: This rule not necessary since *Mathematica* automatically simplifies $\frac{1}{\text{Tan}[z]}$ **to** Cot[z].

Rule:

$$\int \frac{1}{\text{Tan}[c + d\,x]} \, dx \longrightarrow \int \text{Cot}[c + d\,x] \, dx \longrightarrow \frac{\text{Log}[\text{Sin}[c + d\,x]]}{d}$$

Program code:

```
(* Int[1/tan[c_.+d_.*x_],x_Symbol] :=
  Log[RemoveContent[Sin[c+d*x],x]]/d /;
FreeQ[{c,d},x] *)
```

4: $\int (b\,\text{Tan}[c + d\,x])^n \, dx$ when $n \notin \mathbb{Z}$

Derivation: Integration by substitution

Basis: $(b\,\text{Tan}[c + d\,x])^n == \frac{b}{d}\,\text{Subst}\left[\frac{x^n}{b^2+x^2},\, x,\, b\,\text{Tan}[c + d\,x]\right] \partial_x (b\,\text{Tan}[c + d\,x])$

Rule: If $n \notin \mathbb{Z}$**, then**

$$\int (b\,\text{Tan}[c + d\,x])^n \, dx \longrightarrow \frac{b}{d}\,\text{Subst}\left[\int \frac{x^n}{b^2 + x^2} \, dx,\, x,\, b\,\text{Tan}[c + d\,x]\right]$$

Program code:

```
Int[(b_.*tan[c_.+d_.*x_])^n_,x_Symbol] :=
  b/d*Subst[Int[x^n/(b^2+x^2),x],x,b*Tan[c+d*x]] /;
FreeQ[{b,c,d,n},x] && Not[IntegerQ[n]]
```

2. $\int (a + b\,\text{Tan}[c + d\,x])^n \, dx$ when $n \in \mathbb{Z}^+$

1: $\int (a + b\,\text{Tan}[c + d\,x])^2 \, dx$

Derivation: Algebraic expansion

Basis: $(a + b\,\text{Tan}[c + d\,x])^2 == a^2 - b^2 + b^2\,\text{Sec}[c + d\,x]^2 + 2\,a\,b\,\text{Tan}[c + d\,x]$

Rule:

$$\int (a + b\,\mathrm{Tan}[c + d\,x])^2\,dx \longrightarrow \left(a^2 - b^2\right) x + \frac{b^2\,\mathrm{Tan}[c + d\,x]}{d} + 2\,a\,b \int \mathrm{Tan}[c + d\,x]\,dx$$

Program code:

```
Int[(a_+b_.*tan[c_.+d_.*x_])^2,x_Symbol] :=
  (a^2-b^2)*x + b^2*Tan[c+d*x]/d + 2*a*b*Int[Tan[c+d*x],x] /;
FreeQ[{a,b,c,d},x]
```

x: $\int (a + b\,\mathrm{Tan}[c + d\,x])^n\,dx$ when $n \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Note: If common powers of tangents are collected, this results in a compact antiderivative; but requires numerous steps because of fanout.

Rule: If $n \in \mathbb{Z}^+$, then

$$\int (a + b\,\mathrm{Tan}[c + d\,x])^n\,dx \longrightarrow \int \mathrm{ExpandIntegrand}[(a + b\,\mathrm{Tan}[c + d\,x])^n,\ x]\,dx$$

Program code:

```
(* Int[(a_+b_.*tan[c_.+d_.*x_])^n_,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*Tan[c+d*x])^n,x],x] /;
FreeQ[{a,b,c,d},x] && IGtQ[n,0] *)
```

3. $\int (a + b\,\text{Tan}[c + d\,x])^n\,dx$ when $a^2 + b^2 == 0$

1: $\int (a + b\,\text{Tan}[c + d\,x])^n\,dx$ when $a^2 + b^2 == 0 \land n > 1$

Derivation: Symmetric tangent recurrence 1b with $A \to 0,\ B \to 1,\ m \to -1$

Rule: If $a^2 + b^2 == 0 \land n > 1$, then

$$\int (a + b\,\text{Tan}[c + d\,x])^n\,dx \longrightarrow \frac{b\,(a + b\,\text{Tan}[c + d\,x])^{n-1}}{d\,(n - 1)} + 2\,a \int (a + b\,\text{Tan}[c + d\,x])^{n-1}\,dx$$

Program code:

```
Int[(a_+b_.*tan[c_.+d_.*x_])^n_,x_Symbol] :=
  b*(a+b*Tan[c+d*x])^(n-1)/(d*(n-1)) +
  2*a*Int[(a+b*Tan[c+d*x])^(n-1),x] /;
FreeQ[{a,b,c,d},x] && EqQ[a^2+b^2,0] && GtQ[n,1]
```

2: $\int (a + b\,\text{Tan}[c + d\,x])^n\,dx$ when $a^2 + b^2 == 0 \land n < 0$

Derivation: Symmetric tangent recurrence 2a with $A \to 1,\ B \to 0,\ m \to 0$

Rule: If $a^2 + b^2 == 0 \land n < 0$, then

$$\int (a + b\,\text{Tan}[c + d\,x])^n\,dx \longrightarrow \frac{a\,(a + b\,\text{Tan}[c + d\,x])^n}{2\,b\,d\,n} + \frac{1}{2\,a} \int (a + b\,\text{Tan}[c + d\,x])^{n+1}\,dx$$

Program code:

```
Int[(a_+b_.*tan[c_.+d_.*x_])^n_,x_Symbol] :=
  a*(a+b*Tan[c+d*x])^n/(2*b*d*n) +
  1/(2*a)*Int[(a+b*Tan[c+d*x])^(n+1),x] /;
FreeQ[{a,b,c,d},x] && EqQ[a^2+b^2,0] && LtQ[n,0]
```

3: $\int \sqrt{a + b \operatorname{Tan}[c + d x]} \, dx$ when $a^2 + b^2 == 0$

Derivation: Integration by substitution

- **Basis: If** $a^2 + b^2 == 0$, **then** $\sqrt{a + b \operatorname{Tan}[c + d x]} == -\frac{2b}{d} \operatorname{Subst}\left[\frac{1}{2a - x^2}, x, \sqrt{a + b \operatorname{Tan}[c + d x]}\right] \partial_x \sqrt{a + b \operatorname{Tan}[c + d x]}$

- **Rule: If** $a^2 + b^2 == 0$, **then**

$$\int \sqrt{a + b \operatorname{Tan}[c + d x]} \, dx \longrightarrow -\frac{2b}{d} \operatorname{Subst}\left[\int \frac{1}{2a - x^2} \, dx, x, \sqrt{a + b \operatorname{Tan}[c + d x]}\right]$$

- **Program code:**

```
Int[Sqrt[a_+b_.*tan[c_.+d_.*x_]],x_Symbol] :=
  -2*b/d*Subst[Int[1/(2*a-x^2),x],x,Sqrt[a+b*Tan[c+d*x]]] /;
FreeQ[{a,b,c,d},x] && EqQ[a^2+b^2,0]
```

4: $\int (a + b \operatorname{Tan}[c + d x])^n \, dx$ when $a^2 + b^2 == 0$

Derivation: Integration by substitution

- **Basis: If** $a^2 + b^2 == 0$, **then** $(a + b \operatorname{Tan}[c + d x])^n == -\frac{b}{d} \operatorname{Subst}\left[\frac{(a+x)^{n-1}}{a-x}, x, b \operatorname{Tan}[c + d x]\right] \partial_x (b \operatorname{Tan}[c + d x])$

- **Rule: If** $a^2 + b^2 == 0$, **then**

$$\int (a + b \operatorname{Tan}[c + d x])^n \, dx \longrightarrow -\frac{b}{d} \operatorname{Subst}\left[\int \frac{(a + x)^{n-1}}{a - x} \, dx, x, b \operatorname{Tan}[c + d x]\right]$$

- **Program code:**

```
Int[(a_+b_.*tan[c_.+d_.*x_])^n_,x_Symbol] :=
  -b/d*Subst[Int[(a+x)^(n-1)/(a-x),x],x,b*Tan[c+d*x]] /;
FreeQ[{a,b,c,d,n},x] && EqQ[a^2+b^2,0]
```

4. $\int (a + b \operatorname{Tan}[c + d x])^n \, dx$ when $a^2 + b^2 \neq 0$

1: $\int (a + b \operatorname{Tan}[c + d x])^n \, dx$ when $a^2 + b^2 \neq 0 \land n > 1$

Reference: G&R 2.510.1, CRC 423, A&S 4.3.129

Reference: G&R 2.510.4, CRC 427, A&S 4.3.130

Rule: If $a^2 + b^2 \neq 0 \land n > 1$**, then**

$$\int (a + b \operatorname{Tan}[c + d x])^n \, dx \longrightarrow \frac{b (a + b \operatorname{Tan}[c + d x])^{n-1}}{d (n-1)} + \int \left(a^2 - b^2 + 2 a b \operatorname{Tan}[c + d x]\right) (a + b \operatorname{Tan}[c + d x])^{n-2} \, dx$$

Program code:

```
Int[(a_+b_.*tan[c_.+d_.*x_])^n_,x_Symbol] :=
  b*(a+b*Tan[c+d*x])^(n-1)/(d*(n-1)) +
  Int[(a^2-b^2+2*a*b*Tan[c+d*x])*(a+b*Tan[c+d*x])^(n-2),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && GtQ[n,1]
```

2: $\int (a + b \operatorname{Tan}[c + d x])^n \, dx$ when $a^2 + b^2 \neq 0 \land n < -1$

Reference: G&R 2.510.4, CRC 427'

Reference: G&R 2.510.1, CRC 423'

Rule: If $a^2 + b^2 \neq 0 \land n < -1$**, then**

$$\int (a + b \operatorname{Tan}[c + d x])^n \, dx \longrightarrow \frac{b (a + b \operatorname{Tan}[c + d x])^{n+1}}{d (n+1) \left(a^2 + b^2\right)} + \frac{1}{a^2 + b^2} \int (a - b \operatorname{Tan}[c + d x]) (a + b \operatorname{Tan}[c + d x])^{n+1} \, dx$$

Program code:

```
Int[(a_+b_.*tan[c_.+d_.*x_])^n_,x_Symbol] :=
  b*(a+b*Tan[c+d*x])^(n+1)/(d*(n+1)*(a^2+b^2)) +
  1/(a^2+b^2)*Int[(a-b*Tan[c+d*x])*(a+b*Tan[c+d*x])^(n+1),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && LtQ[n,-1]
```

3: $\int \frac{1}{a+b\,\text{Tan}[c+d\,x]}\,dx$ when $a^2+b^2\neq 0$

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{1}{a+b\,z} = \frac{a}{a^2+b^2} + \frac{b\,(b-a\,z)}{(a^2+b^2)\,(a+b\,z)}$
- **Rule: If $a^2+b^2\neq 0$, then**

$$\int \frac{1}{a+b\,\text{Tan}[c+d\,x]}\,dx \longrightarrow \frac{a\,x}{a^2+b^2} + \frac{b}{a^2+b^2}\int \frac{b-a\,\text{Tan}[c+d\,x]}{a+b\,\text{Tan}[c+d\,x]}\,dx$$

- **Program code:**

```
Int[1/(a_+b_.*tan[c_.+d_.*x_]),x_Symbol] :=
  a*x/(a^2+b^2) + b/(a^2+b^2)*Int[(b-a*Tan[c+d*x])/(a+b*Tan[c+d*x]),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0]
```

4: $\int (a+b\,\text{Tan}[c+d\,x])^n\,dx$ when $a^2+b^2\neq 0$

- **Derivation: Integration by substitution**
- **Basis:** $F[b\,\text{Tan}[c+d\,x]] = \frac{b}{d}\,\text{Subst}\left[\frac{F[x]}{b^2+x^2},\ x,\ b\,\text{Tan}[c+d\,x]\right]\,\partial_x\,(b\,\text{Tan}[c+d\,x])$
- **Rule: If $a^2+b^2\neq 0$, then**

$$\int (a+b\,\text{Tan}[c+d\,x])^n\,dx \longrightarrow \frac{b}{d}\,\text{Subst}\left[\int \frac{(a+x)^n}{b^2+x^2}\,dx,\ x,\ b\,\text{Tan}[c+d\,x]\right]$$

- **Program code:**

```
Int[(a_+b_.*tan[c_.+d_.*x_])^n_,x_Symbol] :=
  b/d*Subst[Int[(a+x)^n/(b^2+x^2),x],x,b*Tan[c+d*x]] /;
FreeQ[{a,b,c,d,n},x] && NeQ[a^2+b^2,0]
```